



Συστήματα Γνώσης

Πρακτικό Κομμάτι Μαθήματος
Πρόγραμμα Διάγνωσης Βλάβης βασισμένης σε Μοντέλο

Νίκος Βασιλειάδης, Αναπλ. Καθηγητής
Τμήμα Πληροφορικής



Ευρωπαϊκή Ένωση
Ευρωπαϊκό Κοινωνικό Ταμείο



ΥΠΟΥΡΓΕΙΟ ΠΑΙΔΕΙΑΣ ΚΑΙ ΘΡΗΣΚΕΥΜΑΤΩΝ
ΕΙΔΙΚΗ ΥΠΗΡΕΣΙΑ ΔΙΑΧΕΙΡΙΣΗΣ

Με τη συγχρηματοδότηση της Ελλάδας και της Ευρωπαϊκής Ένωσης



ΕΥΡΩΠΑΪΚΟ ΚΟΙΝΩΝΙΚΟ ΤΑΜΕΙΟ

Άδειες Χρήσης

- Το παρόν εκπαιδευτικό υλικό υπόκειται σε άδειες χρήσης Creative Commons.
- Για εκπαιδευτικό υλικό, όπως εικόνες, που υπόκειται σε άλλου τύπου άδειας χρήσης, η άδεια χρήσης αναφέρεται ρητώς.



Χρηματοδότηση

- Το παρόν εκπαιδευτικό υλικό έχει αναπτυχθεί στα πλαίσια του εκπαιδευτικού έργου του διδάσκοντα.
- Το έργο «Ανοικτά Ακαδημαϊκά Μαθήματα στο Αριστοτέλειο Πανεπιστήμιο Θεσσαλονίκης» έχει χρηματοδοτήσει μόνο τη αναδιαμόρφωση του εκπαιδευτικού υλικού.
- Το έργο υλοποιείται στο πλαίσιο του Επιχειρησιακού Προγράμματος «Εκπαίδευση και Δια Βίου Μάθηση» και συγχρηματοδοτείται από την Ευρωπαϊκή Ένωση (Ευρωπαϊκό Κοινωνικό Ταμείο) και από εθνικούς πόρους.






ΑΡΙΣΤΟΤΕΛΕΙΟ
ΠΑΝΕΠΙΣΤΗΜΙΟ
ΘΕΣΣΑΛΟΝΙΚΗΣ

ΑΝΟΙΚΤΑ
ΑΚΑΔΗΜΑΙΚΑ
ΜΑΘΗΜΑΤΑ



Πρόγραμμα Διάγνωσης Βλάβης βασισμένης σε Μοντέλο

Μοντέλο Μηχανής



Περιγραφή Προβλήματος

- Κάποιοι αισθητήρες παρουσιάζουν ασυμφωνία με τις θεωρητικές τιμές που προβλέπονται από το μοντέλο.
- Στο συγκεκριμένο πρόγραμμα δε μας ενδιαφέρει πώς υπολογίζονται οι θεωρητικές τιμές ή πώς διαπιστώνεται η ασυμφωνία.
- Σκοπός είναι να εντοπιστεί το (ένα και μοναδικό) εξάρτημα που δυσλειτουργεί.
- Υπάρχει περίπτωση να δυσλειτουργεί και ένας ή περισσότεροι αισθητήρες



Εκτέλεση Προγράμματος

```
CLIPS> (reset)
CLIPS> (run)
Which sensors show discrepancy?
([m1] [m2] [m3] [m4] [m5]) m3 m4
Component s23 malfunctions!
CLIPS> (reset)
CLIPS> (run)
Which sensors show discrepancy?
([m1] [m2] [m3] [m4] [m5]) m4 m5
Sensor m4 malfunctions!
Sensor m5 malfunctions!
CLIPS> (reset)
CLIPS> (run)
Which sensors show discrepancy?
([m1] [m2] [m3] [m4] [m5]) m2 m3
Component s12 malfunctions!
```



Κλάση component

```
(defclass component
  (is-a USER)
  (role abstract)
  (slot suspect
    (type SYMBOL)
    (allowed-symbols yes no)
    (default no))
)
```



Κλάση internal-component & sensor

```
(defclass sensor
  (is-a component)
  (role concrete)
  (pattern-match reactive))
```

```
(defclass internal-component
  (is-a component)
  (role concrete)
  (pattern-match reactive)
  (multislot connects_with
    (type INSTANCE-NAME)))
```



Αντικείμενα

```
(definstances sensors-and-internal-components
  (m1 of sensor) (m2 of sensor) (m3 of sensor)
  (m4 of sensor) (m5 of sensor)
  (s14 of internal-component
    (connects with [m5]))
  (s21 of internal-component
    (connects with [m1] [m2]))
  (s22 of internal-component
    (connects with [m2] [m3]))
  (s23 of internal-component
    (connects with [m3] [m4]))
  (s11 of internal-component
    (connects with [s21] [s22]))
  (s12 of internal-component
    (connects with [s22]))
  (s13 of internal-component
    (connects with [s23] [m5])))
```



Αρχικός Κανόνας Εισαγωγής Δεδομένων

```
(defrule initial-question
  ?x <- (initial-fact)
  =>
  (retract ?x)
  (bind $?sensors
    (find-all-instances
      ((?s sensor) TRUE) )
    (set-strategy mea)
    (printout t "Which sensors show
discrepancy? " $?sensors " ")
    (bind $?answer (explode$
      (readline)))
    (assert (discrepancy $?answer))
    (assert (goal make-suspects) )
  )
)
```



Κανόνας Αρχικής Ενοχοποίησης Αισθητήρων

```
(defrule init-suspects
  (goal make-suspects)
  (discrepancy $? ?s $?)
  (object (is-a sensor)
    (name =(symbol-to-instance-name ?s))
    (suspect no))
=>
  (modify-instance
    (symbol-to-instance-name ?s)
    (suspect yes))
)
```




Κανόνας Αρχικής Ενοχοποίησης Αισθητήρων - Εναλλακτικός τρόπος

```
(defrule init-suspects
  (goal make-suspects)
  (discrepancy $? ?s $?)
  (object (is-a sensor)
    (name ?s1 &
      : (eq ?s1 (symbol-to-instance-name ?s))
    )
    (suspect no))
=>
(modify-instance ?s1 (suspect yes))
)
```



Ενοχοποίηση



Κανόνας Ενοχοποίησης Εσωτερικών Εξαρτημάτων – 2 κανόνες

```
(defrule propagate-suspect1
  (goal make-suspects)
  (object (is-a sensor) (name ?s)
           (suspect yes))
  (object (is-a internal-component)
           (name ?c)
           (connects_with $? ?s $?)
           (suspect no))
  =>
  (modify-instance ?c (suspect
yes)))
```



Κανόνας Ενοχοποίησης Εσωτερικών Εξαρτημάτων – 2 κανόνες

```
(defrule propagate-suspect2
  (goal make-suspects)
  (object (is-a internal-component)
    (name ?c1)
    (suspect yes))
  (object (is-a internal-component)
    (name ?c2)
    (connects with $? ?c1 $?)
    (suspect no))
  =>
  (modify-instance ?c2
    (suspect yes)))
```



Κανόνας Ενοχοποίησης Εσωτερικών Εξαρτημάτων – 1 κανόνας

```
(defrule propagate-suspect
  (goal make-suspects)
  (object (is-a component) (name ?c)
           (suspect yes))
  (object (is-a internal-component)
           (name ?c1)
           (connects_with $? ?c $?)
           (suspect no))
  =>
  (modify-instance ?c1 (suspect
yes))
)
```



Κανόνας Αλλαγής Στόχου

```
(defrule make-suspects-continue
```

```
  ?x <- (goal make-suspects)
```

```
=>
```

```
  (retract ?x)
```


```
  (assert
```

```
    (goal exonerate-components) )
```

```
)
```



Σύνδεση με Αθώα Εξαρτήματα



Κανόνας Αθωοποίησης Εξαρτημάτων


Λόγω Σύνδεσης με Αθώα Εξαρτήματα

```
(defrule exonerate-components
  (goal exonerate-components)
  (object (is-a internal-component)
           (name ?c1)
           (suspect yes)
           (connects_with $? ?c $?))
  (object (is-a component)
           (name ?c)
           (suspect no))
  =>
  (modify-instance ?c1 (suspect no))
)
```



Μη-Εξήγηση Ασυμφωνίας

Απλή περίπτωση



Κανόνας Αθωοποίησης Εξαρτημάτων


Λόγω μη-Εξήγησης Ασυμφωνίας (απλός)

```
(defrule exonerate-components-special
  (goal exonerate-components)
  (object (is-a internal-component)
           (name ?c)
           (suspect yes)
           (connects_with ?s))
  (object (is-a sensor) (name ?s)
           (suspect yes))
  (object (is-a sensor) (name ~?s)
           (suspect yes))
  =>
  (modify-instance ?c (suspect no))
)
```



Μη-Εξήγηση Ασυμφωνίας

Γενική περίπτωση



Κανόνας Αθωοποίησης Εξαρτημάτων

Λόγω μη-Εξήγησης Ασυμφωνίας (γενικός)

```
(defrule exonerate-components-special
  (goal exonerate-components)
  (object (is-a internal-component)
    (name ?c)      (suspect yes)
    (connects_with $?s))
  (test (all-inst-of-class sensor $?s))
  (test (all-suspects $?s))
  (object (is-a sensor)
    (name ?s1&:(not (member$ ?s1 $?s)))
    (suspect yes))
  =>
  (modify-instance ?c (suspect no)))
```



Συνάρτηση Ελέγχου Κλάσης

```
(defunction all-inst-of-class
  (?class $?inst)
  (while (> (length$ $?inst) 0) do
    (if (neq (class (nth$ 1 $?inst))
           ?class)
        then return FALSE)
    (bind $?inst (rest$ $?inst)))
  )
(return TRUE)
)
```



Συνάρτηση Ελέγχου «Ενοχής»

```
(defunction all-suspects ($?inst)
  (while (> (length$ $?inst) 0) do
    (if (neq yes
        (send (nth$ 1 $?inst) get-
suspect)
      )
      then return FALSE)
    (bind $?inst (rest$ $?inst))
  )
  (return TRUE)
)
```



Κανόνας Αλλαγής Στόχου

```
(defrule exonerate-components-  
cont
```

```
?x <- (goal exonerate-  
components)
```

```
=>
```


```
(retract ?x)
```

```
(assert (goal announce-result))
```


```
)
```



Αλυσίδα Ενόχων Εξαρτημάτων



Πολλαπλοί Χαλασμένοι Αισθητήρες



Κανόνας Ανακοίνωσης Χαλασμένου Εξαρτήματος

```
(defrule announce-result
  (goal announce-result)
  (object (is-a component) (name ?c)
          (suspect yes))
  (not (object (is-a internal-
component) (name ~?c)
            (suspect yes)
            (connects_with $? ?c $?)))
=>
  (printout t (class ?c) " "
            (instance-name-to-symbol ?c)
            " malfunctions!" crlf)
)
```



Εκτέλεση Προγράμματος

```
CLIPS> (reset)
CLIPS> (run)
Which sensors show discrepancy?
([m1] [m2] [m3] [m4] [m5]) m3 m4
internal-component s23 malfunctions!
CLIPS> (reset)
CLIPS> (run)
Which sensors show discrepancy?
([m1] [m2] [m3] [m4] [m5]) m4 m5
sensor m4 malfunctions!
sensor m5 malfunctions!
CLIPS> (reset)
CLIPS> (run)
Which sensors show discrepancy?
([m1] [m2] [m3] [m4] [m5]) m2 m3
internal-component s12 malfunctions!
```





**Επέκταση Προγράμματος
Διάγνωσης Βλάβης βασισμένης
σε Μοντέλο
Διαπίστωση αρχικής
ασυμφωνίας αισθητήρων**

Εισαγωγή

- Στο προηγούμενο πρόγραμμα θεωρήσαμε ότι η αρχική ασυμφωνία εισάγεται στο πρόγραμμα από τον χρήστη
- Στην πραγματικότητα, το στάδιο παρακολούθησης και εντοπισμού προβλημάτων μπορεί να αυτοματοποιηθεί
- Είσοδος είναι οι τιμές των αισθητήρων και των εντολών




Απλοποιημένο Μοντέλο

- Προς χάριν κατανόησης, το μοντέλο που χρησιμοποιείται στο συγκεκριμένο πρόγραμμα αποτελεί απλούστευση του προηγούμενου
 - Όχι επέκτασή του
- Τα components έχουν μία είσοδο και μία έξοδο μόνο
 - Η συνάρτηση εξόδου υψώνει την είσοδο στο τετράγωνο



Μοντέλο



Κλάσεις

```
(defclass component
  (is-a USER)
  (slot input (type NUMBER))
  (slot output (type NUMBER))
  (multislot connects_with
    (type INSTANCE-NAME) ))

(defclass sensor
  (is-a component)
  (slot reading (type NUMBER)))

(defclass reading
  (is-a USER)
  (multislot values))
```



Στιγμιότυπα

```
(definstances sensors  
  (m1 of sensor))
```

```
(definstances internal-components  
  (c1 of component  
    (connects_with [m1])))
```

```
(definstances readings  
  (r1 of reading  
    (values [c1] 5 [m1] 0)))
```



Μέθοδοι

```
(defmessage-handler component  
  calculate-output primary  
  (?inp)  
    (* ?inp ?inp)  
)
```



Αρχειοποίηση

```
(defrule begin
  ?x <- (initial-fact)
=>
  (retract ?x)
  (set-strategy mea)
  (assert (goal bind-values) )
)
```



Πέρασμα μετρήσεων στο μοντέλο των εξαρτημάτων

```
(defrule bind-values-to-components
```

```
  (goal bind-values)
```

```
  (object (is-a reading) (values $? ?c ?cinp $?))
```

```
  (object (is-a component) (name ?c))
```

=>

```
  (modify-instance ?c (input ?cinp)))
```

```
(defrule bind-values-to-sensors
```

```
  (goal bind-values)
```

```
  (object (is-a reading) (values $? ?s ?read $?))
```

```
  (object (is-a sensor) (name ?s))
```

=>

```
  (modify-instance ?s (reading ?read)))
```



Κανόνας αλλαγής στόχου

```
(defrule change-goal-to-calc-output
```

```
  ?x <- (goal bind-values)
```

```
=>
```

```
  (retract ?x)
```

```
  (assert (goal calc-output))
```

```
)
```



Υπολογισμός «Κανονικών» Τιμών Λειτουργίας των εξαρτημάτων

```
(defrule calc-outputs
  (goal calc-output)
  (object (is-a component)
    (name ?c) (input ?inp-val)))
=>
(modify-instance ?c
  (output (send ?c
    calculate-output ?inp-val))))
```



Κανόνας αλλαγής στόχου

```
(defrule change-goal-to-find-  
discrepancy
```

```
  ?x <- (goal calc-output)
```

```
=>
```

```
  (retract ?x)
```

```
  (assert (goal find-  
discrepancy)))
```



Διαπίστωση (και ανακοίνωση) Ασυμφωνιών

```
(defrule find-discrepancies
```

```
  (goal find-discrepancy)
```

```
  (object (is-a component) (name ?c)
```

```
    (output ?out)
```

```
    (connects_with ?s))
```

```
  (object (is-a sensor) (name ?s)
```

```
    (reading ?sr&~?out))
```

=>

```
(printout t "Sensor " ?s " shows discrepancy: "  
          ?sr " instead of " ?out "!" crlf)
```

```
(assert (discrepancy ?s))
```



Εκτέλεση

```
CLIPS> (reset)
```

```
CLIPS> (run)
```

```
Sensor [m1] shows discrepancy: 0  
instead of 25!
```





Τέλος Ενότητας

Επεξεργασία: Εμμανουήλ Ρήγας

Θεσσαλονίκη, 17/3/2014



Ευρωπαϊκή Ένωση
Ευρωπαϊκό Κοινωνικό Ταμείο



ΥΠΟΥΡΓΕΙΟ ΠΑΙΔΕΙΑΣ & ΘΡΗΣΚΕΥΜΑΤΩΝ, ΠΟΛΙΤΙΣΜΟΥ & ΑΘΛΗΤΙΣΜΟΥ
ΕΙΔΙΚΗ ΥΠΗΡΕΣΙΑ ΔΙΑΧΕΙΡΙΣΗΣ

Με τη συγχρηματοδότηση της Ελλάδας και της Ευρωπαϊκής Ένωσης



ΕΣΠΑ
2007-2013
πρόγραμμα για την ανάπτυξη
ΕΥΡΩΠΑΪΚΟ ΚΟΙΝΩΝΙΚΟ ΤΑΜΕΙΟ